
	Reason for change:
	The ABNF for the P-Charging-Vector access-network-charging-info rule is not alligned with the general ABNF for defining new header field parameters for the P-Charging-Vector header field.

According to RFC 7315, new header field parameters must follow the syntax for generic-param.

P-Charging-Vector  = "P-Charging-Vector" HCOLON icid-value

                                   *(SEMI charge-params)

charge-params        = icid-gen-addr / orig-ioi / term-ioi / transit-ioi / related-icid /

                                   related-icid-gen-addr / generic-param

generic-param         =  token [ EQUAL gen-value ]

gen-value                 =  token / host / quoted-string

According to the syntax above, the value of each header field parameter must follow the syntax for token, host or quoted-string.

However, the header field parameter values for the access-network-charging-info header field parameters do not follow the gen-value syntax.

Currently, as the header field parameter values are not surrounded with quotes, they have to follow the syntax for token. However, many values contain characters that are not covered by token, e.g. commas and semicolons.

Simply placing quotes around the header field parameter values, turning the values into quoted-strings, do not work either, as some of the parameter field parameters values can contain parts that also follow the syntax for quoted-string, e.g. gen-value. It is not possible to have a quoted-string within another quoted-string.

The solution is to turn the header field parameter values into quoted-strings, and to not allow quoted-strings parts within the values. String values are still allowed, but will not be surrounded by quotes, and they must not contain characters which are used to separate value pieces, e.g. comma and semicolon.

	
	

	Summary of change:
	A note is added, indicating that entities must use the syntax above if performing a syntax check of the P-Charging-Vector header field.

	
	

	Consequences if not approved:
	A parser that follows the rules for extending the P-Charging-Vector header field will trigger an error when trying to parse a header field parameter associated with the currently defined access-network-charging-info syntax. This will lead to interoperability problems and failed call establishements.

***** BEGIN MODIFICATION *****
7.2A.5.2.1
General

The syntax of the P-Charging-Vector header field is described in RFC 7315 [52]. There may be additional coding rules for this header field depending on the type of IP-CAN, according to access technology specific descriptions.

Table 7.2A.5 describes 3GPP-specific extensions to the P-Charging-Vector header field defined in RFC 7315 [52].

Table 7.2A.5: Syntax of extensions to P-Charging-Vector header field
   access-network-charging-info = (gprs-charging-info / i-wlan-charging-info / xdsl-charging-info / packetcable-charging-info / icn-charging-info / eps-charging-info / eth-charging-info/ generic-param)

   gprs-charging-info = ggsn SEMI auth-token [SEMI pdp-info-hierarchy] *(SEMI extension-param)

   ggsn = "ggsn" EQUAL gen-value

   pdp-info-hierarchy = "pdp-info" EQUAL LDQUOT pdp-info *(COMMA pdp-info) RDQUOT

   pdp-info = pdp-item SEMI pdp-sig SEMI gcid [SEMI flow-id]

   pdp-item = "pdp-item" EQUAL DIGIT

   pdp-sig = "pdp-sig" EQUAL ("yes" / "no")

   gcid = "gcid" EQUAL 1*HEXDIG
   auth-token = "auth-token" EQUAL 1*HEXDIG

   flow-id = "flow-id" EQUAL "(" "{" 1*DIGIT COMMA 1*DIGIT "}" *(COMMA "{" 1*DIGIT COMMA 1*DIGIT "}")")"

   extension-param = token [EQUAL token]

   i-wlan-charging-info = "pdg"
   xdsl-charging-info = bras SEMI auth-token [SEMI xDSL-bearer-info] *(SEMI extension-param)

   bras = "bras" EQUAL gen-value

   xDSL-bearer-info = "dsl-bearer-info" EQUAL LDQUOT dsl-bearer-info *(COMMA dsl-bearer-info) RDQUOT

   dsl-bearer-info = dsl-bearer-item SEMI dsl-bearer-sig SEMI dslcid [SEMI flow-id]

   dsl-bearer-item = "dsl-bearer-item" EQUAL DIGIT

   dsl-bearer-sig = "dsl-bearer-sig" EQUAL ("yes" / "no")

   dslcid = "dslcid" EQUAL 1*HEXDIG
   packetcable-charging-info = packetcable [SEMI bcid]

   packetcable = "packetcable-multimedia"

   bcid = "bcid" EQUAL 1*48(HEXDIG)

   icn-charging-info = icn-bcp *(SEMI itid) [SEMI extension-param]

   icn-bcp = "icn-bcp" EQUAL gen-value

   itid = itc-sig SEMI itc-id SEMI *(flow-id)

   itc-sig = "itc-sig" EQUAL ("yes" / "no")

   itc-id = "itc-id" EQUAL gen-value

   flow-id = "flow-id" EQUAL gen-value

   extension-param = token [EQUAL (token | quoted-string)]

   eps-charging-info = pdngw [SEMI eps-bearer-hierarchy] *(SEMI extension-param)

   pdngw = "pdngw" EQUAL gen-value

   eps-bearer-hierarchy = "eps-info" EQUAL LDQUOT eps-info *(COMMA eps-info) RDQUOT

   eps-info = eps-item SEMI eps-sig SEMI ecid [SEMI flow-id]

   eps-item = "eps-item" EQUAL DIGIT

   eps-sig = "eps-sig" EQUAL ("yes" / "no")

   ecid = "ecid" EQUAL 1*HEXDIG
   eth-charging-info = ip-edge *(SEMI extension-param)

   fiber-charging-info = ip-edge *(SEMI extension-param)

   ip-edge = "ip-edge" EQUAL gen-value

   extension-param = token [EQUAL (token | quoted-string)]

NOTE:
The syntax above is not aligned with the rules for defining new P-Charging-Vector header field parameters as defined in RFC 7315 [52]. Entities that perform syntax check (even if they are not interested of specific header field parameter values) of the header field need to follow the explicit syntax above, as using the rules in RFC 7315 [52] would trigger a parser error. 
The access-network-charging-info parameter is an instance of generic-param from the current charge-params component of P-Charging-Vector header field.

The access-network-charging-info parameter includes alternative definitions for different types access networks. The description of these parameters are given in the subsequent subclauses.

The "access-network-charging-info" header field parameter is not included in the P-Charging-Vector for SIP signalling that is not associated with a session.

When the "access-network-charging-info" is included in the P-Charging-Vector and necessary information isnot available from the IP-CAN (e.g. via Gx/Rx interface) reference points then null or zero values are included.

For type 1 and type 3 IOIs, the generating SIP entity shall express the "orig-ioi" and "term-ioi" header field parameters in the format of a quoted string as specified in RFC 7315 [52] with a specific string prefix being "Type 1" and "Type 3" respectively to indicate the type of IOI. For the type 2 IOI, no string prefix is used. The receiving SIP entity does not perform syntactic checking of the contents of the IOI parameter (the IOI parameter is passed unmodified to charging entities).

***** END MODIFICATION *****
